NATIONAL RESEARCH COUNCIL OF CANADA
DIVISION OF BUILDING RESEARCH

A FORTRAN IV PROGRAM TO SIMULATE AIR MOVEMENT IN MULTI-STOREY BUILDINGS

by

D. M. Sander and G. T. Tamura

Computer Program No. 35
of the
Division of Building Researc

OTTAWA
March 1973

A FORTRAN IV PROGRAM TO SIMULATE AIR MOVEMENT IN MULTI-STOREY BUILDINGS

by

D.M. Sander and G.T. Tamura

Air leakage through the exterior envelope of a building is an important factor in the calculation of heating and cooling loads, and in connection with humidity and moisture transfer problems. The computations of air flow patterns and rates within a building are important also in relation to the movement of contaminants, such as smoke or odours, and the design of control measures. The purpose of this program is to calculate the air flows and pressure differentials that will occur in a multi-storey building as a result of a combination of wind effect, stack effect and operation of air handling systems.

MATHEMATICAL MODEL OF BUILDING

The building model is an extension of the model previously described by Tamura (1), consisting of leakage openings from outside to each floor area, from each floor area to the floor above, and from each floor area to the vertical shafts. Each shaft may have two vents to outside. These vent openings are designated 'top' and 'bottom' but may be located at any floor level desired. The effects of the air handling systems are incorporated by specifying the net quantity of air supplied to each vertical shaft and each floor area. Stack effect is calculated from the air temperature given at each location. Wind effects are introduced by specifying the pressures due to wind on the left and right sides of the building.

The flow equation used by this program is

$$F = K \Delta P^{x} \qquad (1)$$

where:

F = flow rate (scfm)

K = flow coefficient (scfm/inch w.g.^{x})

x = flow exponent ($0.5 \le x \le 1.0$)

ΔP = pressure differential (inches w.g.)

The flow coefficient, K, is evaluated at standard conditions of pressure and temperature. If the entering air is not at standard temperature a correction is made within the program to account for the effect of viscosity and density. This correction factor is

$$C = \left(\frac{\rho}{\rho_o}\right)^x \left(\frac{\mu}{\mu_o}\right)^{1 - 2x} \quad (2)$$

where:

ρ = density of entering air

ρ_o = density of air at standard temperature

μ = viscosity of entering air

μ_o = viscosity of air at standard temperature

Viscosity and density are calculated from the input temperature.

ASSUMPTIONS AND LIMITATIONS

1) Frictional resistance in vertical shafts is neglected.

2) Net air supplied by air handling system is assumed to be constant and independent of building pressures.

3) The building has an open floor plan with no compartments or vestibules.

4) Pressures, flows and leakage openings are assumed to occur at mid-height of each storey.

DESCRIPTION OF PROGRAM

The set of equations describing the building are obtained by writing mass balance equations for each floor space and each shaft.

For the ith floor,

$$F_{o(i, L)} + F_{o(i, R)} + F_{b(i)} - F_{a(i)} - \sum_{j=1}^{JJ} F_{s(i, j)} + F_{ac(i)} = 0 \quad (3a)$$

and for the jth shaft,

$$\sum_{i=1}^{NN} F_{s(i, j)} + F_{bv(j)} + F_{tv(j)} + F_{sh(j)} = 0 \quad (3b)$$

where:

$F_{o\ (i, L)}$	=	flow from outside left to floor (i)
$F_{o\ (i, R)}$	=	flow from outside right to floor (i)
$F_{b\ (i)}$	=	flow from floor below to floor (i)
$F_{a\ (i)}$	=	flow from floor (i) to floor above
$F_{s\ (i, j)}$	=	flow from floor (i) to shaft (j)
$F_{ac\ (i)}$	=	flow of air supplied to floor (i) by air handling system
$F_{bv\ (j)}$	=	flow into shaft (j) through bottom vent
$F_{tv\ (j)}$	=	flow into shaft (j) through top vent
$F_{sh\ (j)}$	=	flow of outside air supplied into shaft (j) by air handling system
NN	=	number of floors
JJ	=	number of shafts

The flows appearing in equations 3a and 3b are indicated in Figure 1. Combining mass balance equations (3a and 3b) with flow equation (1) results in a set of simultaneous non-linear equations. The outside pressures and the pressure differences due to column weight may be calculated from the input data; the unknown variables are then the floor and shaft pressures. For NN floors and JJ shafts there are NN + JJ equations.

These simultaneous non-linear equations are solved by a method of successive linear approximations. The non-linear function described by equation (1) is shown in Figure 2. In the region near the point, (ΔP_t, F_t), this function may be approximated by a straight line that is tangent to the curve at this point. The equation of this linear function is

$$F = K' [\Delta P - \Delta P_i] \qquad (4a)$$

where

$$K' = K \times \Delta P_t^{\,x-1} \tag{4b}$$

and

$$\Delta P_i = \Delta P_t - \frac{F_t}{K'} \tag{4c}$$

Each leakage element in equations (3a and 3b) may be replaced by this linear approximation. The resulting set of linear equations can then be solved by matrix methods.

The iteration procedure is as follows: an initial linear approximation is made for each element and the resulting equations solved for floor and shaft pressures. The flows corresponding to these pressures are then calculated, and the flow through each element is compared with the flow used for linearization of that element. If the difference is greater than the convergence criterion* that element is re-linearized about the most recently determined flow, and the linear simultaneous equations are solved again. This procedure is repeated until the flow through every element satisfies the convergence criterion. A block diagram of the program is given in Figure 3. A sample input sheet is given in Figure 4.

This program was written for IBM 360 model 67. Execution time using this computer is approximately 9 seconds for a 20-storey building with two shafts, and 90 seconds for a 60-storey building with 7 shafts. The program is dimensioned for a maximum of 100 floors and 10 shafts.

INPUT

Two input routines are provided. The standard input routine expresses leakage coefficients in units of thousands of scfm/(inch w. g.)^x. If input in terms of equivalent orifice areas (sq ft) is desired the optional input routine should be compiled and loaded in place of the standard input routine.

Input is from punched cards as follows:

CARD 1 NN - number of floors (maximum 100)

JJ - number of shafts (maximum 10)

(format 214)

CARDS 2 1 card per floor beginning with floor 1

* The convergence criterion is specified by variable 'error' in sub-routine FUNCT. The value used is .01 pounds/minute (0.13 scfm).

to - height from floor to floor above (ft)

NN+1 - wind pressure left (inches w. g.)

- wind pressure right (inches w. g.)

- net quantity of air supplied to floor (scfm)

(format 3F8. 4, 1F8. 1)

CARD NN+2 - values of flow exponent (left wall, right wall, between floors, and from floor to shafts 1 to JJ)

(format 13F6. 5)

CARDS NN+3 1 card per floor beginning with floor 1

to - values of leakage coefficient (left wall, right

2NN+3 wall, between floors, and from floor to shafts 1 to JJ)

units: $\frac{\text{thousand of scfm}}{(\text{inch w. g.})^x}$

(format 13F6. 2)

CARDS 2NN+4 1 card per shaft beginning with shaft 1

to - net air supplied to shaft (scfm)

2NN+JJ+4 - leakage coefficient of bottom vent, $\frac{\text{thousands scfm}}{(\text{inch w. g.})^x}$

- floor level at which bottom vent located

- side of building on which bottom vent located

(1 - left side, 2 - right side)

- leakage coefficient of top vent, $\frac{\text{thousands scfm}}{(\text{inch w. g.})^x}$

- floor level at which top vent located

- side of building on which top vent located

(1 - left side, 2 - right side)

(format F12.2, 2 (F8.4, 2I4))

CARDS 2NN+JJ+4 1 card per floor beginning with floor 1

to - air temperatures °F. (outside left, outside

3NN+JJ+4 right, floor space, shafts 1 to JJ)

(format 13F6.1)

OUTPUT

A sample of the normal output on the device assigned to IOUT (line printer) is given in Appendix C. In addition the mass flow rates are output on the device assigned to IPUN. This device may be a card punch, magnetic tape unit, or magnetic disc storage. These mass flow rates are intended for later use in a program to calculate smoke concentration. If this output is not desired IPUN should be assigned to a dummy device.

REFERENCE

(1) Tamura, G. T., Computer Analysis of Smoke Movement in Buildings. ASHRAE Transactions, Vol. 75, Part II, 1969, NRCC 11542.



FIGURE 1

AIR FLOWS FOR A TYPICAL FLOOR AND TYPICAL SHAFT

BA 5056-1



FIGURE 2

LINEAR APPROXIMATION OF FLOW EQUATION

BR 5056-2



Fig. 3 (Continued on next page).



Fig. 3 BLOCK DIAGRAM OF PROGRAM

INPUT SHEET

NO. OF FLOORS		NO. OF SHAFTS		(1 CARD)
	4		8	

HEIGHT	WIND PRESS. LEFT	WIND PRESS. RIGHT	A/C P.F. S.C.F.M.
8	16	24	32

(NN CARDS)

VALUE OF X — SHAFTS (1 CARD)

LEFT WALL	RIGHT WALL	BTWN FLOORS	1	2	3	4	5	6	7	8	9	10
6	12	18	24	30	36	42	48	54	60	66	72	78

FLOW COEFFICIENTS OR AREAS (NN CARDS)

SHAFTS

OUT-SIDE LEFT	OUT-SIDE RIGHT	FLOOR ABOVE	1	2	3	4	5	6	7	8	9	10
6	12	18	24	30	36	42	48	54	60	66	72	78

SHAFT VENTS (JJ CARDS)

SHAFT	PRESS. FLOW S.C.F.M.	BOTTOM VENT			TOP VENT		
		SIZE	IBV	JBV	SIZE	ITV	JTV
1							
2							
3							
4							
5							
6							
7							
8							
9							
10							
	12	20	24	28	36	40	44

IBV, ITV - FLOOR ON WHICH VENT IS LOCATED

JBV, JTV - SIDE ON WHICH VENT IS LOCATED - 1 FOR LEFT, 2 FOR RIGHT

TEMPERATURES (NN CARDS)

SHAFTS

OUT-SIDE LEFT	OUT-SIDE RIGHT	FLOOR SPACE	1	2	3	4	5	6	7	8	9	10
6	12	18	24	30	36	42	48	54	60	66	72	78

8R5037

FIGURE 4: Sample Input Sheet

```
C       MAIN PRØGRAM (DMSAIR)
C
C*******************************PURPØSE*****************************
C
C       PRØGRAM TØ CØMPUTE AIR FLØWS AND PRESSURES FØR A BUILDING
C
C       FLØWS AND PRESSURES MAY BE DUE TØ ANY CØMBINATIØN ØF
C                   STACK EFFECT
C                   WIND EFFECT
C                   NET AIR SUPPLIED BY A/C SYSTEM FØR PRESSURIZATIØN
C
C       USES INCLUDE STUDY ØF SMØKE MØVEMENT AND CALCULATIØN ØF AIR INFIL
C
C*******************************************************************
C
C
C
C*******************ASSUMPTIØNS AND LIMITATIØNS*********************
C
C       AIR FLØW EQUATIØN USED IS     F=K*CVISC*(RHØ*DELTA P)**X
C                   WHERE   F= MASS FLØW RATE
C                           K= LEAKAGE CØEFFICIENT
C                           CVISC= CØRRECTIØN FØR VISCØSITY CHANGE
C                           X= LEAKAGE EXPØNENT
C                           RHØ= DENSITY ØF INCØMING AIR
C                           DELTA P= PRESSURE DIFFERENTIAL
C
C       FRICTIØNAL RESISTANCE ØF VERTICAL SHAFTS IS NEGLECTED
C
C       NET AIR FLØW SUPPLIED BY A/C SYSTEM FØR PRESSURIZATIØN IS ASSUMED
C       CØNSTANT  (INDEPENDANT ØF THE FLØØR PRESSURES)
C
C       PRESSURES ARE ASSUMED UNIFØRM THRØUGHØUT THE FLØØR AREA (NØ
C       PRØVISIØN IS MADE FØR VESTIBULES ØR CØMPARTMENTIZATIØN ØF FLØØRS)
C
C       PRØGRAM IS DIMENSIØNED FØR A MAXIMUM ØF 100 FLØØRS AND 10 SHAFTS
C
```

```
C***********************************************************************
C
C
C
C*******************USE WITH THE FOLLOWING SUBROUTINES*****************
C
C                   -INPUT (NOTE....OPTIONAL INPUT AVAILABLE)
C
C                   -INDEP
C
C                   -MATRIX
C
C                   -FUNCT
C
C                   -LINIZE
C
C                   -SIMQ
C
C                   -OUT1
C
C                   -OUT2
C
C***********************************************************************
C
C
C
C*************************** NOTATION *********************************
C
C      PRESSURES:   REFERENCED TO OUTSIDE LEFT AT GROUND LEVEL
C          PRESX(I,1)    -PRESSURE OUTSIDE LEFT WALL AT ITH FLOOR
C          PRESX(I,2)    -PRESSURE OUTSIDE RIGHT WALL AT ITH FLOOR
C          PRESX(I,3)    -PRESSURE DIFFERENCE DUE TO COLUMN HEIGHT
C                         BETWEEN FLOOR I AND I+1
C          PRESX(I,J+3) -PRESSURE DIFFERENCE BETWEEN SHAFT J AT ITH
C                         LEVEL AND BOTTOM OF SHAFT J
C          P(I)          -PRESSURE AT ITH FLOOR LEVEL
C          P(NN+J)       -PRESSURE AT BOTTOM OF SHAFT J
C
```

```
C       FLOWS:
C           F(I,1)          -FLOW FROM OUTSIDE LEFT TO FLOOR I
C           F(I,2)          -FLOW FROM OUTSIDE RIGHT TO FLOOR I
C           F(I,3)          -FLOW FROM FLOOR I TO FLOOR I+1
C           F(I,J+3)        -FLOW FROM FLOOR I TO SHAFT J
C           FBV(J)          -FLOW INTO SHAFT J THROUGH BOTTOM VENT
C
C           FTV(J)          -FLOW INTO SHAFT J THROUGH TOP VENT
C           ACPF(I)         -FLOW SUPPLIED BY A/C SYSTEM TO FLOOR I
C           FANSH           -FLOW SUPPLIED TO SHAFT J BY A/C SYSTEM
C
C       LEAKAGE CONSTANTS:
C           EA(I,1)         -COEFICIENT LEFT WALL AT ITH FLOOR
C           EA(I,2)         -COEFICIENT RIGHT WALL AT ITH FLOOR
C           EA(I,3)         -COEFICIENT FROM FLOOR I TO FLOOR I+1
C           EA(I,J+3)       -COEFICIENT FROM FLOOR I TO SHAFT J
C           TV(J)           -COEFICIENT TOP VENT SHAFT J
C           BV(J)           -COEFICIENT BOTTOM VENT SHAFT J
C           X(J)            -EXPONENT
C
C       INTEGERS:
C
C           NN              -NUMBER OF FLOORS
C           JJ              -NUMBER OF SHAFTS
C           IBV             -LOCATION OF BOTTOM VENT
C           ITV             -LOCATION OF TOP VENT
C           JBV             -1 FOR LEFT SIDE, 2 FOR RIGHT SIDE
C           JTV             -1 FOR LEFT SIDE, 2 FOR RIGHT SIDE
C
C
C
C***************************************************************************
C
C
C
        IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
C
C    ***DIMENSIONS ARE FOR MAXIMUM OF 100 FLOORS AND 10 SHAFTS***
```

```
C
      COMMON/ALL/X(13),NN,JJ,ITERM,IIN,IOUT,IPUN
      COMMON/PARM/H(100),RHO(100,13),EA(100,13),WP(100,2),ACPF(100)
      COMMON/TOPVT/TV(10),FANSH(10),ITV(10),JTV(10)
      COMMON/BOTVT/BV(10),IBV(10),JBV(10)
      COMMON/PRESS/PRESX(100,13),P(110)
      COMMON/LINPAR/G(100,13),PL(100,13),GTV(10),PTVL(10),GBV(10),
     1              PBVL(10)
      COMMON/FLX/F(100,13),FTV(10),FBV(10)
      COMMON/MTX/A(110,110),C(110)
      DIMENSION  AA(12100), PLL(1300)
      EQUIVALENCE (PL(1,1), PLL(1))
      EQUIVALENCE (A(1,1),AA(1))
C
      KS=0
      PO=0.50D0
C
C     .....ASSIGN INPUT/OUTPUT DEVICE NUMBERS.....
C
      IIN=1
      IOUT=3
      IPUN=2
C
    1 CALL INPUT
C
    3 CALL INDEP
C
C     ....INITIAL LINEAR APPROXIMATION....
C     ....STRAIGHT LINE THRU ZERO AND POINT CORRESPONDING TO PINIT....
      NNJJ=NN+JJ
      NNJJ2=NNJJ*NNJJ
      JJ3=JJ+3
      XX=.0750D0**0.50D0
      PPI=PO**(-0.50D0)
      DO 20 J=1,JJ3
C
      DO 10 I=1,NN
      G(I,J)= (EA(I,J)*XX)*PPI
```

```
   10   CONTINUE
   20   CONTINUE
        DO 30 J=1,JJ
        GTV(J)=(TV(J)*XX)*PPI
        GBV(J)=(BV(J)*XX)*PPI
   30   CONTINUE
C      ....SET PL EQUAL TO ZERO....
        DO 40 K=1,1300
        PLL(K)=0.D0
   40   CONTINUE
        DO 50 K=1,10
        PTVL(K)=0.D0
        PBVL(K)=0.D0
   50   CONTINUE
C
        KOUNT=0
C
C     ....ZERO MATRIX A...
        DO 60 K=1,12100
        AA(K)=0.D0
   60   CONTINUE
C
C      ..ZERO MATRIX C...
        DO 65 K=1,110
        C(k)=0.D0
   65   CONTINUE
C
   70   CONTINUE
        CALL MATRIX
C
C       ...COMPRESS MATRIX A INTO AA STORED BY COLUMNS....
        DO 72 J=1,NNJJ
        JAA=J-1
        IAA=JAA*NNJJ
        DO 71 I=1,NNJJ
        IIAA=I+IAA
        AA(IIAA)=A(I,J)
   71   CONTINUE
```

```
   72   CØNTINUE
C
        CALL SIMQ (AA,C,NNJJ,KS)
C       ..PRESSURES P(I) ARE RETURNED IN C...
        IF (KS) 100,73,110
   73   CØNTINUE
        DØ 74 K=1,NNJJ
        P(K)=C(K)
   74   CØNTINUE
C
        CALL FUNCT
C
        KØUNT =KØUNT +1
        IF(ITERM) 80,100,90
   80   STØP
   90   DØ 91 K=1,NNJJ2
        AA(K)=0.0D0
   91   CØNTINUE
        GØ TØ 70
  100   CØNTINUE
C
    4   CALL ØUT1
C
    5   CALL ØUT2
C
    6   WRITE (IØUT,3001)  KØUNT
        WRITE(IØUT,3002)  ITERM
        STØP
C
  110   WRITE (IØUT,3000)
        STØP
 3000   FØRMAT(1X,'SINGULAR MATRIX=NØ SØLUTIØN GIVEN BY SIMQ')
 3001   FØRMAT ('1',20X,'....NUMBER ØF ITERATIØNS =',I4,'.....')
 3002   FØRMAT (////20X,'....NUMBER ØF UNCØNVERGED TERMS=',I4)
        END
```

```
222   FORMAT ('0',7X,'OUTSIDE',3X,'OUTSIDE',3X,
    1         'FLOOR',10(5X,'SHAFT'))
223   FORMAT (3X,13(2X,F8.3))
224   FORMAT (/////40X,'VALUE OF X FOR SHAFT VENTS IS 0.5 ')
230   FORMAT (/////50X,'***** VALUES OF X *****')
300   FORMAT ('1',35X,'(CONTINUED FROM PREVIOUS PAGE)')
      END
```

```
      SUBROUTINE INDEP
C
C     THIS SUBROUTINE IS FOR USE WITH THE AIR FLOW BUILDING MODEL PROGR
C
C     THIS SUBROUTINE CALCULATES ALL INDEPENDANT PRESSURES (THAT IS, TH
C     DUE TO HEIGHT*DENSITY)
C
C
      IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
C
C  ***DIMENSIONS ARE FOR MAXIMUM OF 100 FLOORS AND 10 SHAFTS***
C
      COMMON/ALL/X(13),NN,JJ,ITERM,IIN,IOUT,IPUN
      COMMON/PARM/H(100),RHO(100,13),EA(100,13),WP(100,2),ACPF(100)
      COMMON/TOPVT/TV(10),FANSH(10),ITV(10),JTV(10)
      COMMON/BOTVT/BV(10),IBV(10),JBV(10)
      COMMON/PRESS/PRESX(100,13),P(110)
C
      AVK=.192D0/2.0D0
      NN1=NN-1
C
      DO 3 I=1,NN1
      PRESX(I,3)=(H(I)*RHO(I,3) +H(I+1)*RHO(I+1,3))*AVK
 3    CONTINUE
C
      PRESX(1,1)=0.0D0
      PRESX(1,2)=0.0D0
C
      DO 6 J=1,JJ
      J3=J+3
      PRESX(1,J3)=0.0D0
 6    CONTINUE
C
      DO 4 I=2,NN
      PRESX(I,1)=PRESX(I-1,1)-((H(I)*RHO(I,1)+H(I-1)*RHO(I-1,1))*AVK)
      PRESX(I,2)=PRESX(I-1,2)-((H(I)*RHO(I,2)+H(I-1)*RHO(I-1,2))*AVK)
C
```

```
      DO 5 J=1,JJ
      J3=J+3
      PRESX(I,J3)=PRESX(I-1,J3)-(H(I)*RHO(I,J3)+H(I-1)*RHO(I-1,J3))*AVK
    5 CONTINUE
C
    4 CONTINUE
C
      DO 9 I=1,NN
      PRESX(I,1)=PRESX(I,1)+WP(I,1)
      PRESX(I,2)=PRESX(I,2)+WP(I,2)
    9 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE MATRIX
C
C     THIS SUBROUTINE IS FOR USE WITH THE AIR FLOW BUILDING MODEL PROGR
C
C     THIS SUBROUTINE CREATES THE MATRICES REQUIRED TO SOLVE THE LINEAR
C     SIMULTANEOUS EQUATIONS
C
C
      IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
C
C   ***DIMENSIONS ARE FOR MAXIMUM OF 100 FLOORS AND 10 SHAFTS***
C
      COMMON/ALL/X(13),NN,JJ,ITERM,IIN,IOUT,IPUN
      COMMON/PARM/H(100),RHO(100,13),EA(100,13),WP(100,2),ACPF(100)
      COMMON/TOPVT/TV(10),FANSH(10),ITV(10),JTV(10)
      COMMON/BOTVT/BV(10),IBV(10),JBV(10)
      COMMON/PRESS/PRESX(100,13),P(110)
      COMMON/LINPAR/G(100,13),PL(100,13),GTV(10),PTVL(10),GBV(10),
     1               PBVL(10)
      COMMON/FLX/F(100,13),FTV(10),FBV(10)
      COMMON/MTX/A(110,110),C(110)
C
C    ....CALCULATION OF ELEMENTS OF MATRIX A AND VECTOR C.....
C
      JJ3=JJ+3
      NNJJ=NN+JJ
C
C    ....ELEMENTS OF MATRIX A.....
C
C    ....1ST ROW....
      A(1,1)=0.0D0
      DO 11 J=1,JJ3
      A(1,1) = A(1,1) - G(1,J)
 11   CONTINUE
      A(1,2) = G(1,3)
      DO 21 J=1,JJ
      NNJ=NN+J
```

```
      J3=J+3
      A(1,NNJ) =G(1,J3)
 21   CØNTINUE
C
C     ....RØWS 2 TØ NN-1.......
      NN1=NN-1
      DØ 41 I=2,NN1
      A(I,I-1) = G(I-1,3)
      A(I,I+1) = G(I,3)
      A(I,I) = -G(I-1,3)
      DØ 31 J=1,JJ3
      A(I,I) = A(I,I) - G(I,J)
 31   CØNTINUE
      DØ 51  J=1,JJ
      NNJ=NN+J
      J3=J+3
      A(I,NNJ)=G(I,J3)
 51   CØNTINUE
 41   CØNTINUE
C
C     ....RØW NN......
      A(NN,NN-1) = G(NN-1,3)
      A(NN,NN) = -G(NN-1,3)
      DØ 61 J=1,2
      A(NN,NN) = A(NN,NN) - G(NN,J)
 61   CØNTINUE
      DØ 71 J=4,JJ3
      A(NN,NN) = A (NN,NN) - G(NN,J)
 71   CØNTINUE
      DØ 81  J=1,JJ
      NNJ=NN+J
      J3=J+3
      A(NN,NNJ)=G(NN,J3)
 81   CØNTINUE
C
C     ....RØW NN+J.....
      DØ 101 J=1,JJ
      NNJ=NN+J
```

```
      J3=J+3
      A(NNJ,NNJ)=0.0D0
      DO 91 I=1,NN
      A(NNJ,I)=G(I,J3)
      A(NNJ,NNJ)=A(NNJ,NNJ)-G(I,J3)
 91   CONTINUE
      A(NNJ,NNJ)=A(NNJ,NNJ) -GBV(J) -GTV(J)
101   CONTINUE
C
C    ....ELEMENTS OF VECTOR C....
C
      DO 111 K=1,NN
      C(K)=-(ACPF(K)*.00007500D0)
111   CONTINUE
C
C    ....1ST ROW....
      DO 121 J=1,2
      C(1) = C(1) - G(1,J)*(PRESX(1,J) + PL(1,J))
121   CONTINUE
      DO 131 J=3,JJ3
      C(1) = C(1) - G(1,J) * (PRESX(1,J) - PL(1,J))
131   CONTINUE
C
C    ....ROWS 2 TO NN-1....
      DO 161 I=2,NN1
      DO 141  J=1,2
      C(I)= C(I) - G(I,J) *(PRESX(I,J) + PL(I,J))
141   CONTINUE
      DO 151 J=3,JJ3
      C(I) = C(I) -G(I,J)*(PRESX(I,J) - PL(I,J))
151   CONTINUE
      C(I)=C(I)+G(I-1,3)*(PRESX(I-1,3)-PL(I-1,3))
161   CONTINUE
C
C   .....ROW NN....
      DO 171  J=1,2
      C(NN) =C(NN) - G(NN,J)*(PRESX(NN,J) +PL(NN,J))
171   CONTINUE
```

```
      DØ 181 J=4,JJ3
      C(NN) = C(NN) - G(NN,J)*(PRESX(NN,J) - PL(NN,J))
181   CØNTINUE
      C(NN) = C(NN) +G(NN-1,3) * (PRESX(NN-1,3) - PL(NN-1,3))
C
C    ....RØW NN+J....
      DØ 201 J=1,JJ
      NNJ=NN+J
      IB=IBV(J)
      JB=JBV(J)
      J3=J+3
      C(NNJ)=GBV(J)*(PRESX(IB,J3)-PRESX(IB,JB)-PBVL(J))-(FANSH(J)*
     1         .000075ODO)
      IT=ITV(J)
      JT=JTV(J)
      C(NNJ)=C(NNJ)+GTV(J)*(PRESX(IT,J3)-PRESX(IT,JT)-PTVL(J))
      DØ 191 I=1,NN
      C(NNJ)=C(NNJ)+G(I,J3)*(PRESX(I,J3)-PL(I,J3))
191   CØNTINUE
201   CØNTINUE
C
      RETURN
      END
```

```
      SUBROUTINE SIMQ( A,B,N,KS)
C
C
C     THIS SUBROUTINE IS FOR USE WITH THE AIR FLOW BUILDING MODEL PROGR
C     THIS SUBROUTINE SOLVES THE LINEAR SIMULTANEOUS EQUATIONS
C
      IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
      DIMENSION A(1),B(1)
C     FORWARD SOLUTION
      TOL=0.0D0
      KS=0
      JJ=-N
      DO 65 J=1,N
      JY=J+1
      JJ=JJ+N+1
      BIGA=0.0D0
      IT=JJ-J
      DO 30 I=J,N
C     SEARCH FOR MAXIMUM COEFFICIENT IN COLUMN
      IJ=IT+I
      IF(DABS(BIGA)-DABS(A(IJ))) 20,30,30
   20 BIGA=A(IJ)
      IMAX=I
   30 CONTINUE
C     TEST FOR PIVOT LESS THAN TOLLERANCE (SINGULAR MATRIX)
      IF(DABS(BIGA)-TOL) 35,35,40
   35 KS=1
      RETURN
C     INTERCHANGE ROWS IF NECESSARY
   40 I1=J+N*(J-2)
      IT=IMAX-J
      DO 50 K=J,N
      I1=I1+N
      I2=I1+IT
      SAVE =A(I1)
      A(I1)=A(I2)
      A(I2)=SAVE
```

```
C       DIVIDE EQUATIØN BY LEADING CØEFFICIENT
   50   A(I1)=A(I1)/BIGA
        SAVE=B(IMAX)
        B(IMAX)=B(J)
        B(J)=SAVE/BIGA
C       ELIMINATE NEXT VARIABLE
        IF(J-N) 55,70,55
   55   IQS=N*(J-1)
        DØ 65 IX=JY,N
        IXJ=IQS+IX
        IT=J-IX
        DØ 60 JX=JY,N
        IXJX=N*(JX-1)+IX
        JJX=IXJX+IT
   60   A(IXJX)=A(IXJX)-(A(IXJ)*A(JJX))
   65   B(IX)=B(IX)-(B(J)*A(IXJ))
C       BACK SØLUTIØN
   70   NY=N-1
        IT=N*N
        DØ 80 J=1,NY
        IA=IT-J
        IB=N-J
        IC=N
        DØ 80 K=1,J
        B(IB)=B(IB)-A(IA)*B(IC)
        IA=IA-N
   80   IC=IC-1
        RETURN
        END
```

```
      SUBROUTINE FUNCT
C
C     THIS SUBROUTINE IS FOR USE WITH THE AIR FLOW BUILDING MODEL PROGR
C
C     THIS SUBROUTINE CALCULATES THE FLOW FUNCTION (F) AND CHECKS FOR
C     CONVERGENCE
C
C
      IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
C
C   ***DIMENSIONS ARE FOR MAXIMUM OF 100 FLOORS AND 10 SHAFTS***
C
      COMMON/ALL/X(13),NN,JJ,ITERM,IIN,IOUT,IPUN
      COMMON/PARM/H(100),RHO(100,13),EA(100,13),WP(100,2),ACPF(100)
      COMMON/TOPVT/TV(10),FANSH(10),ITV(10),JTV(10)
      COMMON/BOTVT/BV(10),IBV(10),JBV(10)
      COMMON/PRESS/PRESX(100,13),P(110)
      COMMON/LINPAR/G(100,13),PL(100,13),GTV(10),PTVL(10),GBV(10),
     1              PBVL(10)
      COMMON/FLX/F(100,13),FTV(10),FBV(10)
      COMMON/MTX/A(110,110),C(110)
C
C     .....CALCULATE FLOW FUNCTIONS FOR LINEAR MODEL....
C
      ERROR = .00001
      ITERM = 0
C
      DO 22 J=1,2
      DO 12 I=1,NN
      FX = G(I,J) * (PRESX(I,J)-P(I) + PL(I,J))
      IF (DABS(FX-F(I,J))-ERROR) 12,12,2
    2 F(I,J) =FX
      IF (FX) 4,3,3
    3 DENS = RHO(I,J)
      SIGN=1.
      FXX=FX
      GO TO 5
```

```
    4     DENS = RHO(I,3)
          SIGN=-1.
          FXX=-FX
    5     CONTINUE
          CALL LINIZE (FXX,SIGN,EA(I,J),DENS,G(I,J),PL(I,J),X(J))
          ITERM =ITERM +1
   12     CONTINUE
   22     CONTINUE
C
          NN1=NN-1
C
          DO 42 I=1,NN1
          FX= G(I,3)*(P(I)-P(I+1) +PL(I,3) -PRESX(I,3))
          IF (DABS(FX-F(I,3))-ERROR) 42,42,32
   32     F(I,3) =FX
          IF(FX) 34,33,33
   33     DENS = RHO(I,3)
          SIGN=1.
          FXX=FX
          GO TO 35
   34     DENS = RHO(I+1,3)
          SIGN=-1.
          FXX=-FX
   35     CONTINUE
          CALL LINIZE (FXX,SIGN,EA(I,3),DENS,G(I,3),PL(I,3),X(3))
          ITERM =ITERM +1
   42     CONTINUE
C
          DO 72 J=1,JJ
          NNJ=NN+J
          J3=J+3
          DO 62 I=1,NN
          FX= G(I,J3)*(P(I)-PRESX(I,J3)-P(NNJ) +PL(I,J3))
          IF (DABS(FX-F(I,J3))-ERROR) 62,62,52
   52     F(I,J3) = FX
          IF(FX) 54,53,53
   53     DENS=RHO(I,3)
          SIGN=1.
```

```
      FXX=FX
      GO TO 55
 54   DENS = RHO(I,J3)
      SIGN=-1.
      FXX=-FX
 55   CONTINUE
      CALL LINIZE (FXX,SIGN,EA(I,J3), DENS,G(I,J3),PL(I,J3),X(J3))
      ITERM =ITERM +1
 62   CONTINUE
 72   CONTINUE
C
      XVENT=0.5000
C
      DO 122 J=1,JJ
      NNJ=NN+J
      IB=IBV(J)
      JB=JBV(J)
      IT=ITV(J)
      JT=JTV(J)
      J3=J+3
      FX= GBV(J)*(PRESX(IB,JB) -P(NNJ)-PRESX(IB,J3)+PBVL(J))
      IF (DABS(FX-FBV(J))-ERROR)  92,92,82
 82   FBV(J) = FX
      IF (FX) 84,83,83
 83   DENS=RHO(IB,JB)
      SIGN=1.
      FXX=FX
      GO TO 85
 84   DENS =RHO(IB,J3)
      SIGN=-1.
      FXX=-FX
 85   CONTINUE
      CALL LINIZE (FXX,SIGN,BV(J),DENS,GBV(J),PBVL(J),XVENT)
      ITERM =ITERM +1
 92   CONTINUE
      FX = GTV(J)*(PRESX(IT,JT)-P(NNJ)-PRESX(IT,J3)+PTVL(J))
      IF (DABS(FX-FTV(J))-ERROR) 112,112,102
102   FTV(J) = FX
```

```
      IF (FX) 104,103,103
103   DENS = RHØ(IT,JT)
      SIGN=1.
      FXX=FX
      GØ TØ 105
104   DENS = RHØ(IT,J3)
      SIGN=-1.
      FXX=-FX
105   CØNTINUE
      CALL LINIZE (FXX,SIGN,TV(J),DENS,GTV(J),PTVL(J),XVENT)
      ITERM = ITERM + 1
112   CØNTINUE
122   CØNTINUE
C
      RETURN
          END
```

```
      SUBROUTINE LINIZE(FINIT,SIGN,CONS,DENS,SLOPE,PX,XX)
C
C     THIS SUBROUTINE IS FOR USE WITH THE AIR FLOW BUILDING MODEL PROGR
C
C     THIS SUBROUTINE LINEARIZES THE FLOW EQUATION ABOUT THE OPERATING
C
C
      IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
C
      CONST=CONS
      IF (XX-0.50D0) 3,4,8
C     CALCULATION OF CVISC AND MODIFICATION OF CONST BY CVISC
8     VIREL=(.0750D0/DENS)**.730D0
      CVISC=VIREL**(1.0D0-2.0D0*XX)
      CONST=CONST*CVISC
C
C
C
      IF (XX-1.D0) 4,2,3
C
 4    CONTINUE
      FMIN=.0001
C
      IF (DABS(FINIT)-FMIN) 5,5,1
C
 1    PINIT= (SIGN/DENS)*((FINIT/CONST)**(1.D0/XX))
      SLOPE = XX*CONST*(DENS**XX)*((PINIT*SIGN)**(XX-1.D0))
      PX = SIGN*((FINIT/SLOPE)-(SIGN*PINIT))
      GO TO 6
C
 5    CONTINUE
      PMIN= ((FMIN/CONST)**(1.D0/XX))/DENS
      SLOPE=(1.D0/PMIN)*CONST*((PMIN*DENS)**XX)
      PX=0.
      GO TO 6
C
 2    CONTINUE
```

```
      SLØPE = CØNST*DENS
C
    6 CØNTINUE
      RETURN
C
    3 STØP
      END
```

```
      SUBROUTINE OUT1
C
C     THIS SUBROUTINE IS FOR USE WITH THE AIR FLOW BUILDING MODEL PROGR
C
C     THIS SUBROUTINE CALCULATES AND OUTPUTS PRESSURE DIFFERENCES
C
C
      IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
C
C   ***DIMENSIONS ARE FOR MAXIMUM OF 100 FLOORS AND 10 SHAFTS***
C
      COMMON/ALL/X(13),NN,JJ,ITERM,IIN,IOUT,IPUN
      COMMON/PARM/H(100),RHO(100,13),EA(100,13),WP(100,2),ACPF(100)
      COMMON/TOPVT/TV(10),FANSH(10),ITV(10),JTV(10)
      COMMON/BOTVT/BV(10),IBV(10),JBV(10)
      COMMON/PRESS/PRESX(100,13),P(110)
      COMMON/FLX/F(100,13),FTV(10),FBV(10)
      DIMENSION PPX(10),DELP(100,13),DELPTV(10),DELPBV(10)
      EQUIVALENCE (DELP(1,1),RHO(1,1))
C
      JJ3=JJ+3
      NN1=NN-1
C
C
C     ...CALCULATE PRESSURE DIFFERENCES...
C
      DO 30 J=1,2
      DO 31 I=1,NN
      DELP(I,J)=PRESX(I,J)-P(I)
   31 CONTINUE
   30 CONTINUE
C
      DO 40 I=1,NN1
      DELP(I,3)=P(I+1)-P(I)+PRESX(I,3)
   40 CONTINUE
C
C     ....NO FLOOR ABOVE, SO SET DELP EQUAL TO ZERO....
```

```
      DELP(NN,3)=0.
C
      DO 50 J=1,JJ
      J3=J+3
      NNJ=NN+J
      DO 51 I=1,NN
      DELP(I,J3)=P(NNJ)+PRESX(I,J3)-P(I)
   51 CONTINUE
      ITVJ=ITV(J)
      JTVJ=JTV(J)
      IBVJ=IBV(J)
      JBVJ=JBV(J)
      DELPTV(J)=P(NNJ)+PRESX(ITVJ,J3)-PRESX(ITVJ,JTVJ)
      DELPBV(J)=P(NNJ)+PRESX(IBVJ,J3)-PRESX(IBVJ,JBVJ)
   50 CONTINUE
C
C     ...WRITE PRESSURE DIFFERENCES...
C
      WRITE (IOUT,2000)
      WRITE (IOUT,2001)
      WRITE (IOUT,2002)
      DO 60 I=1,NN
      WRITE (IOUT,101) I,(DELP(I,J),J=1,JJ3)
      IF (I-50) 60,59,60
   59 WRITE (IOUT,300)
      WRITE (IOUT,2001)
      WRITE (IOUT,2002)
   60 CONTINUE
      WRITE (IOUT,2004) (DELPTV(J),J=1,JJ)
      WRITE (IOUT,2005) (DELPBV(J),J=1,JJ)
      WRITE (IOUT,2006)
      RETURN
C
  101 FORMAT (1X, I2 ,13(2X,F8.4))
  300 FORMAT ('1',35X,'(CONTINUED FROM PREVIOUS PAGE)')
 2000 FORMAT ('1',25X,'*****PRESSURE DIFFERENCES (INCHES OF WATER'
     1          ,',REFERENCED TO FLOOR PRESSURE)*****')
 2001 FORMAT ('0','FLOOR',2X,'OUTSIDE',3X,'OUTSIDE',3X,
```

```
     1            'FLOOR',10(5X,'SHAFT'))
2002   FORMAT (10X,'LEFT',5X,'RIGHT',4X,'ABOVE',7X,'1',9X,'2',9X,
     1            '3',9X,'4',9X,'5',9X,'6',9X'7',9X,'8',9X,'9',9X,'10')
2004   FORMAT ( / 13X,'.......TOP VENT.......',10F10.2)
2005   FORMAT (   11X,'.......BOTTOM VENT......',10F10.2)
2006   FORMAT (/////20X,'....VENT PRESSURE DIFFERENCES ARE',
     1            ' RELATIVE TO OUTSIDE AT THAT LEVEL....')
       END
```

```
      SUBROUTINE OUT2
C
C     THIS SUBROUTINE IS FOR USE WITH THE AIR FLOW BUILDING MODEL PROGR
C
C     THIS SUBROUTINE CALCULATES AND OUTPUTS MASS FLOW RATES
C     THIS SUBROUTINE CALCULATES AND OUTPUTS VOLUME FLOW RATES
C
      IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
C
C  ***DIMENSIONS ARE FOR MAXIMUM OF 100 FLOORS AND 10 SHAFTS***
C
      COMMON/ALL/X(13),NN,JJ,ITERM,IIN,IOUT,IPUN
      COMMON/PARM/H(100),RHO(100,13),EA(100,13),WP(100,2),ACPF(100)
      COMMON/BOTVT/BV(10),IBV(10),JBV(10)
      COMMON/TOPVT/TV(10),FANSH(10),ITV(10),JTV(10)
      COMMON/FLX/F(100,13),FTV(10),FBV(10)
      DIMENSION FMASS(100,13),TVMASS(13),BVMASS(13)
      DIMENSION FVOL(100,13),TVVOL(13),BVVOL(13)
      DIMENSION SUMPOS(13), SUMNEG(13)
C
      JJ3=JJ+3
C
C
C
C    ....CALCULATION OF MASS FLOW.....
      DO 130 J=1,JJ3
      SUMPOS(J)=0.0D0
      SUMNEG(J)=0.0D0
      DO 120 I=1,NN
      FMASS(I,J)=F(I,J)*1000.0D0
      IF (FMASS(I,J)) 50,60,60
   50 SUMNEG(J)=SUMNEG(J)+FMASS(I,J)
      GO TO 120
   60 SUMPOS(J)=SUMPOS(J)+FMASS(I,J)
  120 CONTINUE
  130 CONTINUE
      DO 140 J=1,JJ
```

```
      J3=J+3
      TVMASS(J)=FTV(J)*1000.0D0
      IF (TVMASS(J)) 51,61,61
   51 SUMNEG(J3)=SUMNEG(J3)+TVMASS(J)
      GO TO 135
   61 SUMPOS(J3)=SUMPOS(J3)+TVMASS(J)
  135 BVMASS(J)=FBV(J)*1000.0D0
      IF (BVMASS(J)) 52,62,62
   52 SUMNEG(J3)=SUMNEG(J3)+BVMASS(J)
      GO TO 140
   62 SUMPOS(J3)=SUMPOS(J3)+BVMASS(J)
  140 CONTINUE
C
C
C     ....WRITE MASS FLOW RATE....
C     ....WRITE HEADINGS.....
      WRITE (IOUT,1100)
      WRITE (IOUT,1101)
      WRITE (IOUT,1102)
      WRITE (IOUT,1106)
C
      DO 121 I=1,NN
      WRITE (IOUT,1003)I,(FMASS(I,J),J=1,JJ3)
      IF (I-50) 121,21,121
   21 WRITE (IOUT,300)
      WRITE (IOUT,1101)
      WRITE (IOUT,1102)
      WRITE (IOUT,1106)
  121 CONTINUE
      WRITE (IOUT,1004) (TVMASS(J),J=1,JJ)
      WRITE (IOUT,1005) (BVMASS(J),J=1,JJ)
C
C     ...WRITE SUM OF +VE AND -VE FLOWS FOR EACH COLUMN...
      WRITE (IOUT,1107)
      WRITE (IOUT,1105) (SUMPOS(J),J=1,2),(SUMPOS(J),J=4,JJ3)
      WRITE (IOUT,1108) (SUMNEG(J),J=1,2),(SUMNEG(J),J=4,JJ3)
C
      WRITE (IOUT,1109)
```

```
      WRITE (IOUT,1110)
      WRITE (IOUT,1111)
      WRITE (IOUT,1112)
      WRITE (IOUT,1113)
C
C     ......PUNCH OR STORE MASS FLOW DATA FOR LATER USE.....
      DO 122 I=1,NN
      WRITE (IPUN,1012) (FMASS(I,J),J=1,3)
      WRITE (IPUN,1013) (FMASS(I,J),J=4,JJ3)
  122 CONTINUE
      WRITE (IPUN,1013) (TVMASS(J),J=1,JJ)
      WRITE (IPUN,1013) (BVMASS(J),J=1,JJ)
C
C      .....CALCULATION OF VOLUME FLOW RATE.....
      DO 200 J=1,JJ3
      DO 201 I=1,NN
  203 DENS=.07500D0
  205 FVOL(I,J)=FMASS(I,J)/DENS
  201 CONTINUE
      SUMPOS(J)=SUMPOS(J)/DENS
      SUMNEG(J)=SUMNEG(J)/DENS
  200 CONTINUE
      DO 230 J=1,JJ
  235 BVVOL(J)=BVMASS(J)/DENS
  245 TVVOL(J)=TVMASS(J)/DENS
  230 CONTINUE
C
C     ...OUTPUT VOLUME FLOW....
      WRITE (IOUT,1200)
      WRITE (IOUT,1101)
      WRITE (IOUT,1102)
      WRITE (IOUT,1106)
      DO 250 I=1,NN
      WRITE (IOUT,1203) I,(FVOL(I,J),J=1,JJ3)
      IF (I-50) 250,249,250
  249 WRITE (IOUT,300)
      WRITE (IOUT,1101)
      WRITE (IOUT,1102)
```

```
      WRITE (IOUT,1106)
  250 CONTINUE
      WRITE (IOUT,1204) (TVVOL(J),J=1,JJ)
      WRITE (IOUT,1205) (BVVOL(J),J=1,JJ)
C
C     ...WRITE SUM OF +VE AND -VE FLOWS FOR EACH COLUMN...
      WRITE (IOUT,1107)
      WRITE (IOUT,1105) (SUMPOS(J),J=1,2),(SUMPOS(J),J=4,JJ3)
      WRITE (IOUT,1108) (SUMNEG(J),J=1,2),(SUMNEG(J),J=4,JJ3)
C
      WRITE (IOUT,1109)
      WRITE (IOUT,1110)
      WRITE (IOUT,1111)
      WRITE (IOUT,1112)
      WRITE (IOUT,1113)
C
      RETURN
  300 FORMAT ('1',35X,'(CONTINUED FROM PREVIOUS PAGE)')
 1003 FORMAT (1X,I2,13(2X,F8.2))
 1004 FORMAT ( / 13X,'......TOP VENT..(4).',10F10.2)
 1005 FORMAT (    11X,'......BOTTOM VENT.(4).',10F10.2)
 1012 FORMAT (3F8.2)
 1013 FORMAT (10F8.2)
 1100 FORMAT('1',40X,'******MASS FLOW RATE (POUNDS PER MINUTE)******')
 1101 FORMAT ('0','FLOOR',2X,'OUTSIDE',3X,'OUTSIDE',3X,
     1           'FLOOR',10(5X,'SHAFT'))
 1102 FORMAT (10X,'LEFT',5X,'RIGHT',4X,'ABOVE',7X,'1',9X,'2',9X,
     1           '3',9X,'4',9X,'5',9X,'6',9X'7',9X,'8',9X,'9',9X,'10')
 1105 FORMAT (1X,'+VE',F9.0,F10.0,10X,10F10.0)
 1106 FORMAT (10X,'(1)',7X,'(1)',6X,'(2)',10(7X,'(3)'))
 1107 FORMAT (/ 1X,'....SUM OF FLOWS....')
 1108 FORMAT ('0','-VE',F9.0,F10.0,10X,10F10.0)
 1109 FORMAT (///10X,'*****SIGN CONVENTIONS*****')
 1110 FORMAT (10X,'(1)    + INDICATES FLOW FROM OUTSIDE TO FLOOR AREA')
 1111 FORMAT (10X,'(2)    + INDICATES FLOW FROM FLOOR SPACE TO FLOOR'
     1           ,' ABOVE')
 1112 FORMAT (10X,'(3)    + INDICATES FLOW FROM FLOOR SPACE TO SHAFT')
 1113 FORMAT (10X,'(4)    + INDICATES FLOW FROM OUTSIDE TO SHAFT',
```

```
     1            '(THROUGH VENT OPENING)')
1200   FORMAT ('1',45X,'****VOLUME FLOW RATE (SCFM)****')
1203   FORMAT (1X,I2,13(2X,F8.1))
1204   FORMAT ( / 13X,'......TOP VENT..(4).',10F10.1)
1205   FORMAT (    11X,'.....BOTTOM VENT..(4).',10F10.1)
       END
```

```
C
C      INPUT ROUTINE FOR AIR FLOW BUILDING MODEL PROGRAM
C
C      THIS SUBROUTINE  INTENDED FOR  SMOKE MOVEMENT STUDY (INPUT IN
C      EQUIVALENT LEAKAGE AREAS)
C
C      INPUTS ARE AS FOLLOWS...
C                 NN -NUMBER OF FLOORS
C                 JJ -NUMBER OF VERTICAL SHAFTS
C                 H -FLOOR HEIGHTS (FT)
C                 WP -WIND PRESSURES (INCHES OF WATER)
C                 ACPF -NET AIR SUPPLIED BY A/C SYSTEM (SUPPLY-RETURN)
C                       FOR PRESSURIZATION (SCFM)
C                 TEMP -TEMPERATURES (DEGREES F.)
C                 EA -EQUIVILENT LEAKAGE AREAS (SQ.FT.)
C                 X -FLOW EXPONENT
C                 BV -AREA OF BOTTOM VENT OPENING (SQ.FT.)
C                 TV -AREA OF TOP VENT OPENING (SQ.FT.)
C                 ITV -FLOOR LEVEL AT WHICH TOP VENT OPENING OCCURS
C                 IBV -FLOOR LEVEL AT WHICH BOTTOM VENT OPENING OCCURS
C                 JBV -CODE FOR BOTTOM VENT OPENING (1 FOR LEFT SIDE OF
C                      BUILDING, 2 FOR RIGHT SIDE OF BUILDING)
C                 JTV -CODE FOR TOP VENT OPENING (1 FOR LEFT SIDE OF BUILDING
C                      2 FOR RIGHT SIDE OF BUILDING)
C                 FANSH -NET AIR SUPPLIED TO SHAFT FOR PRESSURIZATION
C
C
       IMPLICIT REAL*8(A-H,O-Z),INTEGER*4(I-N)
C
C   ***DIMENSIONS ARE FOR MAXIMUM OF 100 FLOORS AND 10 SHAFTS***
C
       COMMON/ALL/X(13),NN,JJ,ITERM,IIN,IOUT,IPUN
       COMMON/PARM/H(100),RHO(100,13),EA(100,13),WP(100,2),ACPF(100)
       COMMON/TOPVT/TV(10),FANSH(10),ITV(10),JTV(10)
       COMMON/BOTVT/BV(10),IBV(10),JBV(10)
       DIMENSION TEMP(100,13)
       EQUIVALENCE (TEMP(1,1),RHO(1,1))
```

```
C
C      SUBROUTINE TO READ INPUT DATA FOR DMSAIR FROM CARDS OR DATASET
C
       READ (IIN,100) NN,JJ
C
       NNJ=NN+J
       JJ3=JJ+3
C
       READ(IIN,101) (H(I),WP(I,1),WP(I,2),ACPF(I),I=1,NN)
C
       READ(IIN,105)  (X(J),J=1,JJ3)
C
       DO 12 I=1,NN
       READ(IIN,103) (EA(I,J),J=1,JJ3)
  12   CONTINUE
C
       READ (IIN,104)(FANSH(J),BV(J),IBV(J),JBV(J),TV(J),ITV(J),
     1              JTV(J),J=1,JJ)
C
       DO 2 I=1,NN
       READ (IIN,102) (TEMP(I,J),J=1,JJ3)
   2   CONTINUE
C
 100   FORMAT (2I4)
 101   FORMAT (3F8.4,1F8.1)
 102   FORMAT (13F6.1)
 103   FORMAT (13F6.2)
 104   FORMAT (F12.2,F8.4,2I4,F8.4,2I4)
 105   FORMAT (13F6.5)
C
C
C      WRITE INPUT INFORMATION
C
       WRITE (IOUT,200) NN,JJ
C
       WRITE (IOUT,201)
       WRITE (IOUT,204)
       WRITE(IOUT,205)
```

```
      DO 10 I=1,NN
      WRITE (IOUT,206)  I,H(I),WP(I,1),WP(I,2),ACPF(I)
      IF (I-50) 10,11,10
   11 WRITE (IOUT,300)
      WRITE (IOUT,204)
      WRITE (IOUT,205)
   10 CONTINUE
C
      WRITE (IOUT,190)
      WRITE (IOUT,191)
      WRITE (IOUT,192)
      WRITE (IOUT,193)
      WRITE (IOUT,197)
      WRITE (IOUT,194)
      WRITE (IOUT,195)
      WRITE (IOUT,196)
      WRITE (IOUT,230)
      WRITE (IOUT,222)
      WRITE (IOUT,209)
      WRITE (IOUT,223) (X(J) ,J=1,JJ3)
      WRITE (IOUT,224)
C
      WRITE (IOUT,211)
      WRITE (IOUT,212)
      WRITE (IOUT,213)
      DO 18 I=1,NN
      WRITE (IOUT,214) I,(EA(I,J),J=1,JJ3)
      IF (I-50) 18,19,18
   19 WRITE (IOUT,300)
      WRITE (IOUT,212)
      WRITE (IOUT,213)
   18 CONTINUE
C
      WRITE (IOUT,215)
      WRITE (IOUT,216)
      WRITE (IOUT,217)
      DO 60 J=1,JJ
      IF(JBV(J)-1) 20,20,30
```

```
   20    WRITE (IOUT,218) J,FANSH(J),BV(J),IBV(J)
         GO TO 35
   30    WRITE (IOUT,219) J,FANSH(J),BV(J),IBV(J)
   35    IF(JTV(J)-1) 40,40,50
   40    WRITE (IOUT,220) TV(J),ITV(J)
         GO TO 55
   50    WRITE (IOUT,221) TV(J),ITV(J)
   55    CONTINUE
   60    CONTINUE
C
         WRITE (IOUT,207)
         WRITE (IOUT,208)
         WRITE (IOUT,209)
         DO 17 I=1,NN
         WRITE (IOUT,210) I,(TEMP(I,J),J=1,JJ3)
         IF (I-50) 17,16,17
   16    WRITE (IOUT,300)
         WRITE (IOUT,208)
         WRITE (IOUT,209)
   17    CONTINUE
C
C        MULTIPLY LEAKAGE AREAS BY 659 TO GET PROPER UNITS (LB/MIN)
C        DIVIDE LEAKAGE AREAS BY 1000 TO PREVENT EXCESSEVELY LARGE NUMBERS
         DO 5 J=1,JJ3
         DO 4 I=1,NN
         EA(I,J)=EA(I,J)*.6590D0
    4    CONTINUE
    5    CONTINUE
         DO 6 J=1,JJ
         BV(J)=BV(J)*.6590D0
         TV(J)=TV(J)*.6590D0
    6    CONTINUE
C
C        CALCULATE DENSITIES FROM TEMPERATURES
         DO 7 J=1,JJ3
         DO 8 I=1,NN
         RHO(I,J)=39.765780D0/(TEMP(I,J)+459.720D0)
    8    CONTINUE
```

```
7       CONTINUE
C
   70   RETURN
C
C
  190   FORMAT (1H1,35X,'FLOW EQUATION IS...F=659*EA*CVISC*(RHO*',
     1          'DELTA P)**X')
  191   FORMAT ('0',30X,'WHERE:')
  192   FORMAT (35X,'F =MASS FLOW RATE (LB/MIN)')
  193   FORMAT (35X,'EA =EQUIVALENT LEAKAGE AREA (SQ. FT.)')
  194   FORMAT (35X,'RHO =DENSITY OF ENTERING AIR (LB/CU.FT.)')
  195   FORMAT (35X,'DELTA P =PRESSURE DIFFERENTIAL (INCHES OF WATER)')
  196   FORMAT (35X,'X =FLOW EXPONENT')
  197   FORMAT (35X,'CVISC =CORRECTION FOR VISCOSITY CHANGE')
  200   FORMAT('1',40X,'SOLUTION FOR ',I4,' FLOORS, ',I4,' SHAFTS')
  201   FORMAT ('1',40X,'********INPUT PARAMETERS********')
  204   FORMAT (///,1X,'FLOOR',6X,'FLOOR HEIGHT',8X,
     1          'WIND LEFT',9X,'WIND RIGHT',8X,'A/C PRESS. FLOW')
  205   FORMAT (17X,'(FT)',12X,'(INCHES)',11X,'(INCHES)',13X,'(SCFM)')
  206   FORMAT (3X,I3,11X,F5.2,11X,F8.5,11X,F8.5,11X,F8.0)
  207   FORMAT ('1',40X,'*****TEMPERATURES (DEGREES F.)*****')
  208   FORMAT ('0','FLOOR',2X,'OUTSIDE',3X,'OUTSIDE',3X,
     1          'FLOOR',10(5X,'SHAFT'))
  209   FORMAT (10X,'LEFT',5X,'RIGHT',4X,'SPACE',7X,'1',9X,'2',9X,
     1          '3',9X,'4',9X,'5',9X,'6',9X'7',9X,'8',9X,'9',9X,'10')
  210   FORMAT (1X,I2,13(2X,F8.0))
  211   FORMAT ('1',40X,'*****EQUIVALENT LEAKAGE AREAS=SQ.FT.*****')
  212   FORMAT ('0','FLOOR',2X,'OUTSIDE',3X,'OUTSIDE',3X,
     1          'FLOOR',10(5X,'SHAFT'))
  213   FORMAT (10X,'LEFT',5X,'RIGHT',4X,'ABOVE',7X,'1',9X,'2',9X,
     1          '3',9X,'4',9X,'5',9X,'6',9X'7',9X,'8',9X,'9',9X,'10')
  214   FORMAT (1X,I2,13(2X,F8.3))
  215   FORMAT ('1',40X,'*****SHAFT VENTS AND PRESSURIZATION*****')
  216   FORMAT (1H0,1X,'SHAFT',8X,'PRESS FLOW',18X,'****BOTTOM VENT****',
     1          22X,'****TOP VENT*****')
  217   FORMAT (17X,'(SCFM)',2(19X,'SIZE',9X,'LOCATION'))
  218   FORMAT (1X,I4,7X,F12.2,10X,F12.2,8X,I4,' LEFT')
  219   FORMAT (1X,I4,7X,F12.2,10X,F12.2,8X,I4,' RIGHT')
```

```
220   FORMAT ('+',73X,F12.2,8X,I4,' LEFT')
221   FORMAT ('+',73X,F12.2,8X,I4,' RIGHT')
222   FORMAT ('0',7X,'OUTSIDE',3X,'OUTSIDE',3X,
    1         'FLOOR',10(5X,'SHAFT'))
223   FORMAT (3X,13(2X,F8.3))
224   FORMAT (/////40X,'VALUE OF X FOR SHAFT VENTS IS 0.5 ')
230   FORMAT (/////50X,'***** VALUES OF X *****')
300   FORMAT ('1',35X,'(CONTINUED FROM PREVIOUS PAGE)')
      END
```

```
!EOD
REC=  1293, FILES=    1
!EOD

ET=004.68
03/01/73  1051   BK=004.70,FG=000.02,ID=000.00

!FIN
```

APPENDIX B - SAMPLE INPUT DATA

```
 15      2
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
12.         0.        0.          200.
.66      .66     .5      .5      .5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
  4.0      4.0     9.5     9.5     2.5
           0.0        0.0     1      1    20.0     15    1
           0.0        0.0     1      1      .0     15    1
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
    .0       .0    75.     75.     75.     75.     75.     75.
```

APPENDIX C - SAMPLE OUTPUT

SOLUTION FOR 15 FLOORS,   2 SHAFTS

********INPUT PARAMETERS********

FLOOR	FLOOR HEIGHT (FT)	WIND LEFT (INCHES)	WIND RIGHT (INCHES)	A/C PRESS. FLOW (SCFM)
1	12.00	0.00000	0.00000	200.
2	12.00	0.00000	0.00000	200.
3	12.00	0.00000	0.00000	200.
4	12.00	0.00000	0.00000	200.
5	12.00	0.00000	0.00000	200.
6	12.00	0.00000	0.00000	200.
7	12.00	0.00000	0.00000	200.
8	12.00	0.00000	0.00000	200.
9	12.00	0.00000	0.00000	200.
10	12.00	0.00000	0.00000	200.
11	12.00	0.00000	0.00000	200.
12	12.00	0.00000	0.00000	200.
13	12.00	0.00000	0.00000	200.
14	12.00	0.00000	0.00000	200.
15	12.00	0.00000	0.00000	200.

FLOW EQUATION IS...F=COEF*CVISC*(RHO*DELTA P)**X

WHERE:

F - MASS FLOW RATE (LB/MIN)
COEF - LEAKAGE COEFFICIENT (1000 LB. PER MIN/INCH)
CVISC - CORRECTION FOR VISCOSITY CHANGE
RHO - DENSITY OF ENTERING AIR (LB/CU. FT.)
DELTA P - PRESSURE DIFFERENTIAL (INCHES OF WATER)
X - FLOW EXPONENT

***** VALUES OF X *****

OUTSIDE LEFT	OUTSIDE RIGHT	FLOOR SPACE	SHAFT 1	SHAFT 2	SHAFT 3	SHAFT 4	SHAFT 5	SHAFT 6
0.660	0.660	0.500	0.500	0.500				

VALUE OF X FOR SHAFT VENTS IS 0.5

****LEAKAGE COEFFICIENTS-(1000 SCFM/INCH)****

FLOOR	OUTSIDE LEFT	OUTSIDE RIGHT	FLOOR ABOVE	SHAFT 1	SHAFT 2	SHAFT 3	SHAFT 4	SHAFT 5	SHAFT 6
1	4.000	4.000	9.500	9.500	2.500				
2	4.000	4.000	9.500	9.500	2.500				
3	4.000	4.000	9.500	9.500	2.500				
4	4.000	4.000	9.500	9.500	2.500				
5	4.000	4.000	9.500	9.500	2.500				
6	4.000	4.000	9.500	9.500	2.500				
7	4.000	4.000	9.500	9.500	2.500				
8	4.000	4.000	9.500	9.500	2.500				
9	4.000	4.000	9.500	9.500	2.500				
10	4.000	4.000	9.500	9.500	2.500				
11	4.000	4.000	9.500	9.500	2.500				
12	4.000	4.000	9.500	9.500	2.500				
13	4.000	4.000	9.500	9.500	2.500				
14	4.000	4.000	9.500	9.500	2.500				
15	4.000	4.000	9.500	9.500	2.500				

*****SHAFT VENTS AND PRESSURIZATION*****

SHAFT	PRESS FLOW (SCFM)	****BOTTOM VENT**** SIZE	LOCATION	*****TOP VENT***** SIZE	LOCATION
1	0.00	0.0	1 LEFT	20.00	15 LEFT
2	0.00	0.0	1 LEFT	0.00	15 LEFT

*****TEMPERATURES (DEGREES F.)*****

FLOOR	OUTSIDE LEFT	OUTSIDE RIGHT	FLOOR SPACE	SHAFT 1	SHAFT 2	SHAFT 3	SHAFT 4	SHAFT 5	SHAFT 6
1	0.	0.	75.	75.	75.				
2	0.	0.	75.	75.	75.				
3	0.	0.	75.	75.	75.				
4	0.	0.	75.	75.	75.				
5	0.	0.	75.	75.	75.				
6	0.	0.	75.	75.	75.				
7	0.	0.	75.	75.	75.				
8	0.	0.	75.	75.	75.				
9	0.	0.	75.	75.	75.				
10	0.	0.	75.	75.	75.				
11	0.	0.	75.	75.	75.				
12	0.	0.	75.	75.	75.				
13	0.	0.	75.	75.	75.				
14	0.	0.	75.	75.	75.				
15	0.	0.	75.	75.	75.				

*****PRESSURE DIFFERENCES (INCHES OF WATER, REFERENCED TO FLOOR PRESSURE)*****

FLOOR	OUTSIDE LEFT	OUTSIDE RIGHT	FLOOR ABOVE	SHAFT 1	SHAFT 2	SHAFT 3	SHAFT 4	SHAFT 5	SHAFT 6
1	0.1730	0.1730	-0.0022	-0.0501	-0.0386				
2	0.1473	0.1473	-0.0050	-0.0479	-0.0363				
3	0.1243	0.1243	-0.0066	-0.0429	-0.0313				
4	0.1029	0.1029	-0.0071	-0.0363	-0.0248				
5	0.0821	0.0821	-0.0071	-0.0292	-0.0176				
6	0.0613	0.0613	-0.0067	-0.0221	-0.0105				
7	0.0400	0.0400	-0.0060	-0.0154	-0.0038				
8	0.0181	0.0181	-0.0068	-0.0094	0.0022				
9	-0.0031	-0.0031	-0.0035	-0.0026	0.0090				
10	-0.0276	-0.0276	-0.0037	0.0009	0.0125				
11	-0.0518	-0.0518	-0.0042	0.0046	0.0162				
12	-0.0756	-0.0756	-0.0042	0.0088	0.0204				
13	-0.0993	-0.0993	-0.0034	0.0130	0.0245				
14	-0.1239	-0.1239	-0.0016	0.0164	0.0279				
15	-0.1503	-0.1503	0.0000	0.0180	0.0295				
		......TOP VENT......		0.17	0.18				
		......BOTTOM VENT......		-0.22	-0.21				

....VENT PRESSURE DIFFERENCES ARE RELATIVE TO OUTSIDE AT THAT LEVEL....

******MASS FLOW RATE (POUNDS PER MINUTE)******

FLOOR	OUTSIDE LEFT (1)	OUTSIDE RIGHT (1)	FLOOR ABOVE (2)	SHAFT 1 (3)	SHAFT 2 (3)	SHAFT 3 (3)	SHAFT 4 (3)	SHAFT 5 (3)	SHAFT 6 (3)
1	107.05	107.05	33.58	158.85	36.67				
2	96.26	96.26	50.25	155.26	35.59				
3	86.08	86.08	57.47	146.90	33.04				
4	76.00	76.00	59.89	135.20	29.38				
5	65.46	65.46	59.80	121.21	24.80				
6	53.96	53.96	58.11	105.44	19.16				
7	40.74	40.74	55.06	87.98	11.55				
8	24.12	24.12	58.40	68.63	-8.75				
9	-6.57	-6.57	41.87	36.06	-17.69				
10	-27.83	-27.83	43.31	-21.27	-20.83				
11	-42.20	-42.20	45.91	-48.25	-23.75				
12	-54.14	-54.14	45.88	-66.60	-26.64				
13	-64.85	-64.85	41.31	-80.88	-29.25				
14	-75.03	-75.03	28.26	-90.81	-31.21				
15	-85.22	-85.22	0.00	-95.11	-32.08				
		......TOP VENT..(4).		-612.64	0.00				
		......BOTTOM VENT.(4).		0.00	0.00				
....SUM OF FLOWS....									
+VE	550.	550.		1016.	190.				
-VE	-356.	-356.		-1016.	-190.				

*****SIGN CONVENTIONS*****

(1) + INDICATES FLOW FROM OUTSIDE TO FLOOR AREA

(2) + INDICATES FLOW FROM FLOOR SPACE TO FLOOR ABOVE

(3) + INDICATES FLOW FROM FLOOR SPACE TO SHAFT

(4) + INDICATES FLOW FROM OUTSIDE TO SHAFT (THROUGH VENT OPENING)

****VOLUME FLOW RATE (SCFM)****

FLOOR	OUTSIDE LEFT	OUTSIDE RIGHT	FLOOR ABOVE	SHAFT 1 (3)	SHAFT 2 (3)	SHAFT 3 (3)	SHAFT 4 (3)	SHAFT 5 (3)	SHAFT 6 (3)
1	1427.3	1427.3	447.7	2118.0	488.9				
2	1283.5	1283.5	670.0	2070.1	474.5				
3	1147.8	1147.8	766.3	1958.7	440.6				
4	1013.3	1013.3	798.5	1802.6	391.7				
5	872.8	872.8	797.3	1616.2	330.6				
6	719.4	719.4	774.8	1405.8	255.5				
7	543.1	543.1	734.1	1173.1	154.0				
8	321.5	321.5	778.7	915.1	-116.7				
9	-87.6	-87.6	558.2	480.9	-235.8				
10	-371.0	-371.0	577.4	-283.6	-277.8				
11	-562.7	-562.7	612.2	-643.3	-316.6				
12	-721.8	-721.8	611.7	-888.0	-355.2				
13	-864.7	-864.7	550.7	-1078.4	-390.0				
14	-1000.4	-1000.4	376.8	-1210.9	-416.1				
15	-1136.3	-1136.3	0.0	-1268.2	-427.7				
	......TOP VENT..(4).			-8168.5	0.0				
	.....BOTTOM VENT..(4).			0.0	0.0				
....SUM OF FLOWS....									
+VE	7329.	7329.		13540.	2536.				
-VE	-4745.	-4745.		-13541.	-2536.				

*****SIGN CONVENTIONS*****

(1) + INDICATES FLOW FROM OUTSIDE TO FLOOR AREA
(2) + INDICATES FLOW FROM FLOOR SPACE TO FLOOR ABOVE
(3) + INDICATES FLOW FROM FLOOR SPACE TO SHAFT
(4) + INDICATES FLOW FROM OUTSIDE TO SHAFT (THROUGH VENT OPENING)

....NUMBER OF ITERATIONS = 7.....

....NUMBER OF UNCONVERGED TERMS = 0